<commit_before>| manifest load |

"Disable underscore as assignment, allowing underscores in method names."
Scanner allowUnderscoreAsAssignment: false.

manifest := #(	('http://source.squeak.org/FFI'					1	('FFI-Pools' 'FFI-Kernel'))
				('http://source.squeak.org/VMMaker'				6	('CogPools-ISAs' 'BytecodeSets.spur' 'VMMaker.oscog' 'VMMakerUI' 'Cog' 'CogTools' 'ImageFormat' 'FileAttributesPlugin.oscog'))
				('http://ss3.gemstone.com/ss/MethodMassage'		3	('MethodMassage' 'MethodMassageCompatibility'))
				('http://www.squeaksource.com/AioPlugin'		7	('VMConstruction-Plugins-AioPlugin.oscog'))
				('http://www.squeaksource.com/Alien'			0	('Alien-Core' 'Alien-Tests-Core' 'Alien-Win32'))
				"('http://www.squeaksource.com/FreeTypePlus'	5	('FreeType')) can't load this.  it is toxic to Squeak 4.5"
				"('http://www.squeaksource.com/FreetypePlugin'	8	('Freetype-Plugin')) hence we can do without this"
				('http://www.squeaksource.com/OSProcess'		4	('OSProcess'))
				('http://www.squeaksource.com/OSProcessPlugin'	9	('VMConstruction-Plugins-OSProcessPlugin.oscog'))
				('http://www.squeaksource.com/rb'				5	('AST-Core' 'AST-Semantic' 'AST-Tests-Core' 'AST-Tests-Semantic' 'Refactoring-Changes' 'Refactoring-Core' 'Refactoring-Environment' 'Refactoring-Tests-Changes' 'Refactoring-Tests-Core' 'Refactoring-Tests-Environment' 'Refactoring-Squeak-Platform'))
				"This for the Klatt speech synthesiser, but seems to be AWOL"
				('http://www.squeaksource.com/Speech'			2	('SharedPool-Speech' ))
				('http://www.squeaksource.com/XDCP'				9	('VMConstruction-Plugins-XDisplayControlPlugin.oscog' ))
				('http://www.squeaksource.com/Balloon3D'		9	('Balloon3D-Constants' 'Balloon3D-Plugins'  ))
				('http://www.squeaksource.com/Cryptography'		9	('CryptographyPlugins'  ))
				('http://ss3.gemstone.com/ss/AndreasSystemProfiler'	9	('AndreasProfiler'))
				('http://www.squeaksource.com/Printf'			5	('Printf'))
				('http://www.squeaksource.com/VectorEnginePlugin'       9       ('VectorEnginePlugin'))
				).

load := (manifest collect:
				[:tuple|
				[:path :order :packages| | repository |
				repository := MCHttpRepository
								location: path
								user: 'squeak'
								password: '<PASSWORD>'.
				MCRepositoryGroup default addRepository: repository.
				{repository. order. packages}] valueWithArguments: tuple])
			sort: [:a :b| a second <= b second].

#(	"'FT2Constants.st'" ) do:
	[:fileName| (FileDirectory default fileNamed: fileName) fileIn].

load do:
	[:tuple|
	 [:repository :order :packages|
	  packages do:
		[:package | | packageVersions latestVersion |
		"We need to filter-out branches of unbranched packages, but can't assume
		the package list is (reverse) ordered."
		packageVersions := ((repository versionNamesForPackageNamed: package)
			select: [:versionName| (versionName at: package size + 1) = $-])
			asSortedCollection: [ :a :b | a versionNumber < b versionNumber ].
		latestVersion := packageVersions last.
		[| version |
       		 version := (MCCacheRepository default versionNamed: latestVersion)
				ifNil: [repository versionNamed: latestVersion].
		 version load.
		 version workingCopy repositoryGroup addRepository: repository]
			on: Warning
			do: [:ex|
				((ex messageText beginsWith: 'This package depends on the following classes')
				 or: [ex messageText beginsWith: 'About to serialize an empty diffy version.']) ifFalse:
					[ex pass].
				ex resume]]]
		valueWithArguments: tuple].

World findATranscript: nil.

UIManager default
	edit: (FileDirectory default fileNamed: 'Workspace.text') contentsOfEntireFile
	label: 'Overview'.


#('VM Simulation' 'Source Generation' 'Slang Test' 'In-image Compilation') reverseDo:
	[:nickname| | label workspace |
	label := nickname, ' Workspace'.
	workspace := (UIManager default
					edit: (FileDirectory default fileNamed: label, '.text') contentsOfEntireFile
					label: label) model.
	workspace shouldStyle ifFalse:
		[workspace toggleStyling]].

(Utilities classPool at: #CommonRequestStrings) ifNotNil:
	[:crs|
	crs contents: crs contents, '\-\VMMaker generateConfiguration\VMMaker generateAllConfigurationsUnderVersionControl\VMClass openCogMultiWindowBrowser\VMClass openObjectMemoriesInterpretersBrowser\VMClass openSpurMultiWindowBrowser' withCRs].

CodeHolder useMultiWindowBrowsers: true.

#('Cog' 'CogTools' 'VMMaker') do:
	[:pkg|
	(PackageInfo named: pkg) classes do:
		[:c|
		c organization sortCategories.
		c class organization sortCategories]].
(Smalltalk classNamed: #StackInterpreter) withAllSubclasses do:
	[:sic| sic reorganizeAsISeeFit].

Smalltalk snapshot: true andQuit: true<commit_msg>Minor clean-up in image-preparation script for recent Squeak Trunk (i.e., Squeak6.0alpha). It now matches what will be fetched through the VMMaker update map, which currently includes the code for plugins Freetype, DBus, Rome, and GStreamer -- without the image code.<commit_after>| manifest load |

"Update Trunk image"
[MCMcmUpdater default doUpdate: false] on: Warning do: [:ex | ex resume: true].

"Disable underscore as assignment, allowing underscores in method names."
Scanner allowUnderscoreAsAssignment: false.

manifest := #(	
	squeak	'FFI'
		0	('FFI-Pools' 'FFI-Kernel')
	squeak	'FFI' "Split up because of how extension methods are loaded"
		1	('FFI-Libraries' 'FFI-Tools' 'FFI-Callbacks'
			'FFI-Tests' 'FFI-PoolsTests' 'FFI-CallbacksTests')
	squeak	'VMMaker'
		6	('CogPools-ISAs' 'BytecodeSets.spur' 'VMMaker.oscog' 'VMMakerUI'
			'Cog' 'CogTools' 'ImageFormat' 'FileAttributesPlugin.oscog')
	squeaksource3	'MethodMassage'
		3	('MethodMassage' 'MethodMassageCompatibility')
	squeaksource		'AioPlugin'
		7	('VMConstruction-Plugins-AioPlugin.oscog')
	squeaksource		'Alien'			
		0	('Alien-Core' 'Alien-Tests-Core' 'Alien-Win32')
	squeaksource3	'FreeTypePlus' "more recent than on squeaksource"
		5	('FreeTypeConstants' "'FreeType'")
	squeaksource		'FreetypePlugin'
		8	('Freetype-Plugin')
	squeaksource		'OSProcess'
		4	('OSProcess')
	squeaksource		'OSProcessPlugin'
		9	('VMConstruction-Plugins-OSProcessPlugin.oscog')
	squeaksource		'rb'
		5	('AST-Core' 'AST-Semantic' 'AST-Tests-Core' 'AST-Tests-Semantic'
			'Refactoring-Changes' 'Refactoring-Core' 'Refactoring-Environment'
			'Refactoring-Tests-Changes' 'Refactoring-Tests-Core'
			'Refactoring-Tests-Environment' 'Refactoring-Squeak-Platform'
			'Refactoring-Critics' 'Refactoring-Spelling')
	squeaksource		'Speech' "This for the Klatt speech synthesiser, but seems to be AWOL"
		2	('SharedPool-Speech' )
	squeaksource		'XDCP'
		9	('VMConstruction-Plugins-XDisplayControlPlugin.oscog' )
	squeaksource		'Balloon3D'
		9	('Balloon3D-Constants' 'Balloon3D-Plugins'  )
	squeaksource		'Cryptography'
		9	('CryptographyPlugins'  )
	squeaksource3	'AndreasSystemProfiler'
		9	('AndreasProfiler')
	squeaksource		'Printf'
		5	('Printf')
	squeaksource		'VectorEnginePlugin'
		9	('VectorEnginePlugin')
	squeaksource		'dbus'
		9	('DBus-Plugin')
	squeaksource		'Rome'
		9	('Rome-Base' 'Rome-PluginCanvas' 'Rome-Plugin')
	squeaksource		'GStreamer'
		9	('GStreamer-Plugins')
	).

load := Array streamContents: [:s |
	manifest groupsDo: [:installerSymbol :project :order :packages |
		| installer |
		installer := Installer perform: installerSymbol.
		installer project: project.
		installer packages addAll: packages.
		s nextPut: order -> installer]].
load := (load sorted: [:a :b | a key <= b key]) collect: [:ea | ea value].
load do: [:installer | installer installQuietly].

self currentWorld findATranscript: nil.

Project uiManager
	edit: (FileDirectory default fileNamed: 'Workspace.text') contentsOfEntireFile
	label: 'Overview'
	shouldStyle: false.

#('VM Simulation' 'Source Generation' 'Slang Test' 'In-image Compilation')
	reverseDo: [:nickname| | label |
	label := nickname, ' Workspace'.
	Project uiManager
		edit: (FileDirectory default fileNamed: label, '.text') contentsOfEntireFile
		label: label
		shouldStyle: true].

Utilities appendToCommonRequests: '-\VMMaker generateConfiguration\VMMaker generateAllConfigurationsUnderVersionControl\VMClass openCogMultiWindowBrowser\VMClass openObjectMemoriesInterpretersBrowser\VMClass openSpurMultiWindowBrowser' withCRs.

CodeHolder useMultiWindowBrowsers: true.

#('Cog' 'CogTools' 'VMMaker') do:
	[:pkg|
	(PackageInfo named: pkg) classes do:
		[:c|
		c organization sortCategories.
		c class organization sortCategories]].

(Smalltalk classNamed: #StackInterpreter) withAllSubclasses do:
	[:sic| sic reorganizeAsISeeFit].

Smalltalk snapshot: true andQuit: true
{'fraction_non_alphanumeric': 0.12667923638934717, 'fraction_numerical': 0.0069526278576478905, 'mean_word_length': 4.947442186405046, 'pattern_counts': {'":': 0, '<': 7, '<?xml version=': 0, '>': 5, 'https://': 0, 'lorem ipsum': 0, 'www.': 13, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2388810', 'n_tokens_mistral': 3129, 'n_tokens_neox': 2889, 'n_words': 688}